Assignment 4
Solution Guideline

From the data set `assign4.dta`:

The study on bankruptcy firm employs the following regression model.

$$z_i = \beta_0 + \beta_1 x_{1i} + \beta_2 x_{2i} + \beta_3 x_{3i} + \beta_4 x_{4i} \tag{1}$$

The log-likelihood function of this model is as follows:

$$\ln L = \begin{cases} \ln \Phi(z_i) \ if \ \ y_i = 1 \\ \ln \Phi(-z_i) \ if \ \ y_i = 0 \end{cases} \tag{2}$$

where: y_i = 1 for bankruptcy firm and 0 otherwise.

x_{1i} = Debt coverage ratio of firm i

x_{2i} = Liquidity ratio of firm i

x_{3i} = Profitability index of firm i

x_{4i} = Solidity ratio of firm i

Let $\Phi(.)$ = Logistic probability distribution function. $\Phi(z_i) = \dfrac{1}{1+e^{-z_i}}$

From the given data set (`assig8-2.dta`):

a. Estimate the above models using MLE with Newton-Ralphson algorithm.

Do-file

```
program ml_logit
   args lnf theta
   quietly replace `lnf'=ln(1/(1+exp(-`theta'))) if $ML_y1==1
   quietly replace `lnf'=ln(1-(1/(1+exp(-`theta')))) if $ML_y1==0
end
```

```
. do "D:\EE625-626\Assignment\2020\assign8 Q2 - ml_logit.do"

. program ml_logit
  1.    args lnf theta
  2.    quietly replace `lnf'=ln(1/(1+exp(-`theta'))) if $ML_y1==1
  3.    quietly replace `lnf'=ln(1-(1/(1+exp(-`theta')))) if $ML_y1==0
  4. end

.
end of do-file

. ml model lf (y=x1 x2 x3 x4)
invalid something: unmatched close parenthesis or bracket

r(198);

. ml model lf ml_logit (y=x1 x2 x3 x4)

. ml maximize

initial:       log likelihood = -90.109133
alternative:   log likelihood = -86.130008
rescale:       log likelihood = -86.130008
Iteration 0:   log likelihood = -86.130008
```

```
...
Iteration 7:   log likelihood = -54.627603

                                                Number of obs     =        130
                                                Wald chi2(4)      =      22.79
Log likelihood = -54.627603                     Prob > chi2       =     0.0001

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |   .4835128   .1686119     2.87   0.004     .1530395     .813986
          x2 |   1.454009   .5001373     2.91   0.004     .4737577     2.43426
          x3 |   2.173186   .7757021     2.80   0.005      .652838    3.693535
          x4 |   1.855464   .7138855     2.60   0.009     .4562738    3.254653
       _cons |  -1.400447   .5531237    -2.53   0.011    -2.484549    -.316344
------------------------------------------------------------------------------
```

b. Perform hypothesis testing whether $\beta_1 = \beta_2 = \beta_3 = \beta_4 = 0$ using LR-test and Wald test.

Wald Test

```
. test x1 x2 x3 x4

 ( 1)  [eq1]x1 = 0
 ( 2)  [eq1]x2 = 0
 ( 3)  [eq1]x3 = 0
 ( 4)  [eq1]x4 = 0

           chi2(  4) =   22.79
         Prob > chi2 =    0.0001

. est store ur

. ml model lf ml_logit (y=)

. ml maximize

initial:       log likelihood = -90.109133
alternative:   log likelihood = -86.130008
rescale:       log likelihood = -86.130008
Iteration 0:   log likelihood = -86.130008
Iteration 1:   log likelihood = -86.129902
Iteration 2:   log likelihood = -86.129902

                                                Number of obs     =        130
                                                Wald chi2(0)      =          .
Log likelihood = -86.129902                     Prob > chi2       =          .

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
       _cons |   .5026289   .1809802     2.78   0.005     .1479141    .8573436
------------------------------------------------------------------------------

. est store res
```

LR Test

```
. lrtest ur res

Likelihood-ratio test                                 LR chi2(4)  =     63.00
(Assumption: res nested in ur)                        Prob > chi2 =    0.0000
```

c. Estimate the above models using MLE with BHHH algorithm, make comparison of the estimated result with the result from (1), and give explanation why are they different?

BHHH Algorithm

```
. ml model lf ml_logit (y=x*), tech(bhhh)

. ml maximize

initial:       log likelihood = -90.109133
alternative:   log likelihood = -86.130008
rescale:       log likelihood = -86.130008
Iteration 0:   log likelihood = -86.130008
...
Iteration 25:  log likelihood = -54.627605

                                                  Number of obs   =        130
                                                  Wald chi2(4)    =      16.34
Log likelihood = -54.627605                       Prob > chi2     =     0.0026

------------------------------------------------------------------------------
             |                 OPG
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |   .4833318   .1542635     3.13   0.002       .180981    .7856827
          x2 |   1.453275    .597925     2.43   0.015      .2813639    2.625187
          x3 |   2.172641   .8589781     2.53   0.011       .489075    3.856207
          x4 |   1.854961   .7142729     2.60   0.009      .4550117     3.25491
       _cons |  -1.400063   .5625155    -2.49   0.013     -2.502573   -.2975527
------------------------------------------------------------------------------

. est store bhhh

. est table unres bhhh, star(.1 .05 .01) stat(N ll chi2 p)
estimation result unres not found
r(111);

. est table ur bhhh, star(.1 .05 .01) stat(N ll chi2 p)

----------------------------------------------
    Variable |      ur            bhhh
-------------+--------------------------------
          x1 |  .48351277***    .48333183***
          x2 |  1.4540089***    1.4532755**
          x3 |  2.1731863***    2.1726412**
          x4 |  1.8554636***    1.8549608***
       _cons | -1.4004466**    -1.4000628**
-------------+--------------------------------
           N |        130             130
          ll | -54.627603      -54.627605
        chi2 |  22.786172       16.338035
           p |  .00013971       .00259755
----------------------------------------------
           legend: * p<.1; ** p<.05; *** p<.01
```

The estimated results using different algorithm can be different because of the different computation function.

Let $\Phi(.)$ = Cumulation standard normal probability distribution function and

$$z_i = \beta_0 + \beta_1 x_{1i} + \beta_2 x_{2i} \tag{3}$$

d. Estimate the models using MLE with Newton-Ralphson algorithm.

Do-file

```
program ml_probit
   args lnf theta
   tempvar z
   quietly g double `z'=`theta'
   quietly replace `lnf'=ln(normal(`z')) if $ML_y1==1
   quietly replace `lnf'=ln(1-normal(`z')) if $ML_y1==0
end
```

```
. do "D:\EE625-626\Assignment\2020\assign8 Q2 - ml_probit.do"

. program ml_probit
  1.    args lnf theta
  2.    tempvar z
  3.    quietly g double `z'=`theta'
  4.    quietly replace `lnf'=ln(normal(`z')) if $ML_y1==1
  5.    quietly replace `lnf'=ln(1-normal(`z')) if $ML_y1==0
  6. end

.
end of do-file

. ml model lf ml_probit (y=x1 x2)

. ml maximize

initial:       log likelihood = -90.109133
alternative:   log likelihood = -87.504336
rescale:       log likelihood = -86.291737
Iteration 0:   log likelihood = -86.291737
...
Iteration 6:   log likelihood = -60.695503

                                                Number of obs     =        130
                                                Wald chi2(2)      =      22.61
Log likelihood = -60.695503                     Prob > chi2       =     0.0000

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
          x1 |   .3592444   .0871351     4.12   0.000     .1884627    .5300262
          x2 |   1.139607   .3224845     3.53   0.000     .5075491    1.771665
       _cons |  -.7729211   .2765064    -2.80   0.005    -1.314864   -.2309785
------------------------------------------------------------------------------

. est store probit
```

Assume that there exists heteroskedasticity in the model as: $\sigma_i^2 = \exp(\gamma x_{4i})^2$, then,
$\Phi(.)$ = Cumulation standard normal probability distribution function $\Phi\ z_i / \exp(\gamma x_{4i})$

e. Estimate the models with heteroskedasticity using MLE with Newton-Ralphson algorithm. Perform LR-test whether there exists significant heteroskedasticity.

Do-file

```
program ml_probit_het
   args lnf theta sigma
   tempvar z s
   quietly g double `s'=exp(`sigma')
   quietly g double `z'=`theta'/`s'
   quietly replace `lnf'=ln(normal(`z')) if $ML_y1==1
   quietly replace `lnf'=ln(1-normal(`z')) if $ML_y1==0
end
```

```
. do "D:\EE625-626\Assignment\2020\assign8 Q2 - ml_probit_het.do"

. program ml_probit_het
  1.    args lnf theta sigma
  2.    tempvar z s
  3.    quietly g double `s'=exp(`sigma')
  4.    quietly g double `z'=`theta'/`s'
  5.    quietly replace `lnf'=ln(normal(`z')) if $ML_y1==1
  6.    quietly replace `lnf'=ln(1-normal(`z')) if $ML_y1==0
  7. end

.
end of do-file

. ml model lf ml_probit_het (y=x1 x2) (x4, noconstant)

. ml maximize

initial:       log likelihood = -90.109133
alternative:   log likelihood = -83.932039
rescale:       log likelihood = -83.932039
rescale eq:    log likelihood =  -70.42663
Iteration 0:   log likelihood =  -70.42663
...
Iteration 11:  log likelihood = -59.404451

                                                Number of obs     =        130
                                                Wald chi2(2)      =      16.42
Log likelihood = -59.404451                     Prob > chi2       =     0.0003

------------------------------------------------------------------------------
           y |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
eq1          |
          x1 |   .2921402   .0773775     3.78   0.000      .140483    .4437974
          x2 |   .9417315   .2800569     3.36   0.001     .3928299    1.490633
       _cons |  -.6388427   .2144265    -2.98   0.003    -1.059111   -.2185744
-------------+----------------------------------------------------------------
eq2          |
          x4 |    1.18393    .585199     2.02   0.043     .0369609    2.330899
------------------------------------------------------------------------------
```

```
. est store hetprob

. lrtest hetprob probit

Likelihood-ratio test                                 LR chi2(1)  =      2.58
(Assumption: probit nested in hetprob)                Prob > chi2 =    0.1081

. est table probit hetprob, star(.1 .05 .01) stat(N ll chi2 p)

----------------------------------------------
    Variable |    probit          hetprob
-------------+--------------------------------
eq1          |
          x1 |  .35924444***    .29214017***
          x2 |  1.1396071***    .94173148***
       _cons | -.77292114***   -.63884268***
-------------+--------------------------------
eq2          |
          x4 |                  1.1839299**
-------------+--------------------------------
Statistics   |
           N |        130             130
          ll | -60.695503      -59.404451
        chi2 |  22.614244       16.415643
           p |  .00001229       .00027251
----------------------------------------------
           legend: * p<.1; ** p<.05; *** p<.01
```

Since p-value of LR-test is greater than 0.05, the null hypothesis of no heteroskedasticity is not rejected. Thus, there is insignificant heteroskedasticity even though the estimated coefficient in variance equation is significant.